
/**
 This example shows how we can read in data from a CSV file and 
 create a basic visualization. 
 
 In this case, we read in some US census data from 2009 and create
 a scatter plot showing average houshold income versus the percentage
 of the population with a higher degree.
 
 C. Andrews
 2014-01-21
 **/

// The Table class is the heart of the way we manage data
Table table;
void setup() {
  size(800, 800);
  // This reads the csv file and creates a Table object to hold the data
  table = loadTable("census.csv", "header");

  // there is no need for animation, so we’ll turn on noLoop()
  noLoop();

  // we want to display the names of the states, so we will load a font
  PFont font = createFont("Arial", 9, true);
  textFont(font);
  textAlign(CENTER);
}

/**
 The drawing is fairly straightforward. We iterate 
 through each row of the table, extracting the information we want.
 We use map() to transform the income and degree data into x,y
 coordinates, and then we just draw a circle there. Finally, we
 write the state name centered over the top of each circle.
 
 It is not very pretty (the names all overlap), but it works.
 **/
void draw() {
  background(255);
  // new for loop syntax
  // this runs the loop once for each item in the list of the right,
  // assigning it into the variable on the left
  for (TableRow row: table.rows()) {

    // we use the getFloat(), getInt() and getString() methods of
    // the row to get the data out of the row in the right format
    int income = row.getInt("income");
    float degree = row.getFloat("% degree");
    String state = row.getString("State");

    // map the income and degree to x,y coordinates on the canvas
    // note the flip in the y dimension to ivnert the y axis
    float x = map(income, 30000, 70000, 50, width);
    float y = map(degree, 17, 50, height - 50, 50);

    // draw the circle at the correct spot
    fill(255);
    ellipse(x, y, 10, 10);

    // draw the label over the circle
    fill(0);
    text(state, x, y - 10);
  }
}


